
Delphi How To
第5章 網際網路應用篇

第1章
Resource File

1-1 資源檔(.res)的產生

使用元件 

無

原理解說 

Resource是指可以嵌入Win32程式的二進位資料, 它可以是Windows標準的resource, 也可以由使用者自訂(application-defined); 標準的resource像是Icon, Cursor, Bitmap, Font, String或Version Information等。

產生 .res檔的方法有下列幾種:

1.使用resource script(.rc檔)與BRCC32.Exe

2.使用Borland Resource Workshop、Image Editor這類的resource editor.

3.直接依resource的格式將其resources寫入檔案內

這裏我們說明第一種方法.

所謂resource script(.rc檔)是文字檔, 其內容列出有哪些resources, 格式為ResName + ResType + ResFileName, 如下所示:

ResName1  ResType1  ResFileName1
ResName2  ResType2  ResFileName2 
... 
ResName欄位指定一個識別resource的唯一名稱或整數值;  ResType欄位指resource的型態, resource型態列舉如下:

	RT_ACCELERATOR
	Accelerator table

	RT_BITMAP
	Bitmap resource

	RT_DIALOG
	Dialog box

	RT_FONT
	Font resource

	RT_FONTDIR
	Font directory resource

	RT_MENU
	Menu resource

	RT_RCDATA
	Application-defined resource(raw data)

	RT_STRING
	String-table entry

	RT_MESSAGETABLE
	Message-table entry

	RT_CURSOR
	Hardware-dependent cursor resource

	RT_GROUP_CURSOR
	Hardware-independent cursor resource

	RT_ICON
	Hardware-dependend icon resource

	RT_GROUP_ICON
	Hardware-independent icon resource

	RT_VERSION
	Version resource


ResFileName欄位則指resource的檔案名稱, 若該檔案與 .rc檔不在同一路徑的話, 則須加上檔案的完整路徑.

另外, 多種resource可以放在同一個資源檔中, 將resource加入應用程序中, 使用時比resource在硬碟中的花費時間要少,所以應用程序執行速度會較快, 但產生的應用程序的檔案size較大。

設計步驟 

1.產生一個副檔名為rc的空白檔案

2.依上述格式填入resource

3.利用Program Files\Borland\Delphi5\Bin目錄下BRCC32.Exe產生 .res檔

4.以{$R 資源檔案名.RES}的方式, 將resource連結至Delphi專案中

範例說明 

產生一個名為sample.rc的文字檔, 存在C:\Test目錄下, sample.rc檔的內容如下:

1  BITMAP  "C:\Test\1.bmp"
MyBmp  BITMAP  "C:\WINNT\Winnt256.bmp"
MyJpeg  JPEG  beauty.jpg

1  ICON  dialog.ico

car  ICON  car.ico

pikacur  CURSOR  pikacur.cur

MyAni1 ANICURSOR  globe.ani
MyAni2 ANICURSOR  m_ball.ani
dunk  AVI  dunk.avi
start  WAVE  "C:\WINNT\Media\start.wav"
Lsansuni  MYFONT  "C:\Windows\Font\Lsansuni.ttf"
TestHtml  23  test1.html

HtmlGif  2110  4in1.gif
接著進入 MS-Dos 模式, 準備Compile剛完成的sample.rc檔. 假設Delphi安裝於C:\Program Files目錄, 執行Command如下:

C:\Test>C:\Program Files\Borland\Delphi5\Bin\BRCC32  sample.rc

執行完畢, 在C:\Test的目錄下即產生一個sample.RES檔。

在Delphi專案中的Main Form的Unit找到 {$R *.DFM} 這行, 在這行的下方加上 {$R sample.res} 或 {$RESOURCE sample.res}, 這樣即可將sample.res這個resource file載入Delphi的專案中。

$R是Compiler Directives之一, 其格式為 {$R 資源檔案名稱} 或 {$Resource 資源檔案名稱}。這個編譯指令的用意在指定要含入指定的資源檔. 這資源檔必須是標準的Windows resource file, 且副檔名是 .RES. 如果檔案名稱中有空白的話, 須用單引號, 例: {$R ‘My resource’}

技巧提示

.rc的主檔名不可與Delphi的 任一個unit檔與project檔相同名稱, 因為Delphi會為Project及每個unit產生對應的 .res檔。

1-2 列舉一module內所有resource
使用元件 

TButton, TTreeView

原理解說 

利用Windows提供的EnumResourceNames、EnumResourceTypes、EnumResourceLanguages與EnumResTypeProc四個API來列舉一個module內各種resource的相關資訊。這裡以EnumResourceNames這個API為範例, 以一個callback function, 將一個module內的各resource的Name與Type顯示在TTreeView元件上。
設計步驟 

1.定義EnumResourceNames API的Callback function: EnumResNameProc。所謂Callback function是指將一個procedure或 function當作一參數, 傳給另一個procedure, 所以當這個procedure被呼叫時, 這個callback function也會被呼叫, 等到callback function執行完畢時, 主控權便傳回原來的procedure。
2.在btnEnumResourceNames的OnClick以LoadLibrary/LoadLibraryEx將可執行的module(如Exe, DLL)載入﹐並將它映射到目前進程使用的位址空間。一旦載入﹐即可存取庫內儲存的資源。
3.執行EnumResourceNames API的動作。

範例說明 

var

  LastNode: TTreeNode;

  LastType: string;
const // 定義各種Resource Type的陣列

  RESSTRINGS: array[1..22] of string = ('RT_CURSOR', 'RT_BITMAP', 'RT_ICON', 'RT_MENU', 'RT_DIALOG', 'RT_STRING', 'RT_FONTDIR', 'RT_FONT', 'RT_ACCELERATOR', 'RT_RCDATA', 'RT_MESSAGETABLE',                'RT_GROUP_CURSOR', 'RT_13', 'RT_GROUP_ICON', 'RT_15',                'RT_VERSION', 'RT_DLGINCLUDE', 'RT_18', 'RT_PLUGPLAY',                'RT_VXD', 'RT_ANICURSOR', 'RT_ANIICON');

// EnumResNameProc為EnumResourceNames所須的Callback function

function EnumResNameProc(hModule: THandle; lpszType, lpszName: PChar; lParam: LongInt): Boolean; stdcall; // 記得最後加上 stdcall

var

  sType: string;

  tvItem: TTreeNode;

begin

  if (LongInt(lpszType) > 256) then

    sType := String(lpszType)

  else

    sType := RESSTRINGS[LongInt(lpszType)];
  if sType <> LastType then

  begin

    LastNode := Form1.TreeView1.Items.Add(nil, sType);

    LastType := sType;

  end;
  if LongInt(lpszName) < 65535 then

    tvItem := Form1.TreeView1.Items.AddChild(LastNode, IntToStr(LongInt(lpszName)))

  else

    tvItem := Form1.TreeView1.Items.AddChild(LastNode, string(lpszName));

  end;
  // 必須回傳True, 才能繼續Enumerate的動作, 回傳False則停止動作

  Result := True; 

end;

procedure TForm1.btnEnumResourceNamesClick(Sender: TObject);

var

  hMod: Thandle;

  resType: Byte;

begin

  LastType := '';

  TreeView1.Items.Clear;
  // 取得控制程式本身module的Handle

  hMod := LoadLibraryEx(ParamStr(0)), 0, LOAD_LIBRARY_AS_DATAFILE);

  if hMod = 0 then // LoadLibraryEx動作失敗

    exit

  else

  try

    for resType := 1 to 22 do

      EnumResourceNames(hMod, MakeIntResource(resType), @EnumResNameProc, 0);

  finally

    FreeLibrary(hMod);

  end;

end;

技巧提示
以下為EnumResNameProc的宣告:

BOOL CALLBACK EnumResNameProc(
    HANDLE hModule,
// resource-module handle 

    LPCTSTR lpszType,
// pointer to resource type 

    LPTSTR lpszName,
// pointer to resource name 

    LONG lParam 
// application-defined parameter  

   );
hModule參數指要列舉resource的module的handle, 由LoadLibrary或LoadLibraryEx API的回傳值取得。若此參數為0, 則會列舉產生目前這個processs的module內的resource names。

lpzsType參數指列舉得之resource的type字串值的指標。

lpszName參數指列舉得之resource的name字串值的指標。

lParam參數為程式自訂的參數。

以下為EnumResourceNames API的宣告:

BOOL EnumResourceNames(

    HINSTANCE hModule,
// resource-module handling 

    LPCTSTR lpszType,
// pointer to resource type 

    ENUMRESNAMEPROC lpEnumFunc,
// pointer to callback function 

    LONG lParam 
// application-defined parameter 

   );
lpEnumFunc參數為一callback function的指標, 其型態為ENUMRESNAMEPROC, 在範例中宣告了一個EnumResNameProc的callback function, 使用時以 @EnumResNameProc取得callback function的指標位址, 並傳給EnumResourceNames的第三個參數。

MakeIntResource宣告如下:

LPTSTR MAKEINTRESOURCE(

    WORD wInteger
// integer to convert 

   );
MakeIntResource API可將一整數轉換成PChar型態。

1-3 使用resource中的Bitmap
使用元件 

TButton, TImage

原理解說 

利用TBitmap的LoadFromResourceName與LoadFromResourceID兩個method, 以及LoadBmp API來取得resource中的Bitmap。

設計步驟 

1.將Bitmap的resource連結至專案中

2.第一個範例利用TBitmap的LoadFromResourceName method, 將指定的resource name 的Bitmap自resource中取出, 載入Image1元件中

3.第二個範例利用TBitmap的LoadFromResourceID method, 將指定的resource ID 的Bitmap自resource中取出, 載入Image1元件中

4.第三個範例利用LoadBitmap API取得resource中的Bitmap

範例說明 

範例一:

LoadFromResourceName的宣告如下, 

procedure LoadFromResourceName(Instance: THandle; const ResName: string);
Instance參數指要取得的resource的module handle, 範例的hInstance指Application的Instanece。

ResName參數則指resource中的Bitmap的識別名稱。

本例將ResName為MyBmp的Bitmap載入Image1中。

procedure TForm1.Button1Click(Sender: TObject);

begin
  Image1.Picture.Bitmap.LoadFromResourceName(hInstance, 'MyBmp');

end;
範例二:

LoadFromResourceID的宣告如下,

procedure LoadFromResourceID(Instance: THandle; ResID: Integer);
ResID參數指resource中Bitmap的識別ID, 為整數型態。

本例將resource中ResID為1的Bitmap畫在Image1的Canvas上。
var

 bmp: TBitmap;

begin

  bmp := TBitmap.Create;
  try

    Bmp.LoadFromResourceID(hInstance, 1);

    Image1.Canvas.Draw(0, 0, bmp);
  finally

    bmp.Free;
  end;

end;

範例三:
LoadBitmap的宣告如下,

HBITMAP LoadBitmap(

    HINSTANCE hInstance,
// handle of application instance 

    LPCTSTR lpBitmapName
// address of bitmap resource name 

   );
lpBitmapName參數指resource中的Bitmap識別名稱。

若載入resource成功, 則回傳Bitmap類別的Handle, 失敗則回傳值為0。

(另外Help中說明, 以TBitmap.LoadFromResourceName的方式取代LoadBitmap API; 且LoadBitmap不支援256色的image)

var

 bmp: TBitmap;

begin

  bmp := TBitmap.Create;
  try

    // 將LoadBitmap所得之Handle指定給bmp.Handle

bmp.Handle := LoadBitmap(hInstance, PChar('MyBmp'));

// 若bmp.Handle值為0, 表示LoadBitmap失敗
    if bmp.Handle <> 0 then
      Image1.Picture.Bmp.Assign(bmp);

  finally

    bmp.Free;
  end;

end;

1-4 使用resource中的Jpeg
使用元件 

TImage

原理解說 

設計步驟 

1.將Jpeg的resource連結至專案中

2.範例一使用FindResource/LoadResource/LockResource/SizeOfResource等API與一TMemoryStream, 將resource中的JPEG顯示在Image1上。

3.範例二使用TResourceStream將resource中的JPEG顯示在Image1上。

範例說明 

範例一:

FindResource的宣告如下,

HRSRC FindResource(
    HMODULE hModule, // resource-module handle 

    LPCTSTR lpName,
// pointer to resource name  

    LPCTSTR lpType 
// pointer to resource type 

   );
hModule參數為module的handle。lpName與lpType參數則分別指要找的resource的名稱與型態的指標。若函式失敗則回傳值為0, 若函式成功則回傳值為指定的resource資訊區塊的handle, 將此handle傳給LoadResource API的第二個參數, 以取得指定resource的handle。

LoadResource的宣告如下,

HGLOBAL LoadResource(

    HMODULE hModule,
// resource-module handle  

    HRSRC hResInfo 
// resource handle 

   );
LoadResource將指定的resource載入記憶體中, 成功則回傳值為指定的resource在記憶內的handle, 失敗則回傳值為0。

LockResource的宣告如下,

LPVOID LockResource(

    HGLOBAL hResData 
// handle to resource to lock 

   );
LockResource鎖住在記憶體中的特定memory。

SizeofResource的宣告如下,

DWORD SizeofResource(

    HMODULE hModule,
// resource-module handle  

    HRSRC hResInfo 
// resource handle 

   );
以LoadResource取得的resource handle當SizeofResource的第二個參數, 取得指定resource佔了多少byte數。

uses Jpeg;

procedure TForm1.btnJpegFromResourceClick(Sender: TObject);
var

  hRes: THandle;

  hMem: THandle;

  MemStream: TMemoryStream;

  pRes: PByte;

  ResSize: Longint;

  JPEGImage: TJPEGImage;

begin

  MemStream := TMemoryStream.Create;

  JPEGImage := TJPEGImage.Create;
Try

  // 取得ResName為Kaoru在hInstace中的Handle

hRes := FindResource(hInstance, PChar('Kaoru'), 'JPEG');
// 將hRes這個handle所指的resource載入memory

hMem := LoadResource(hInstance, hRes);
// 將hMem在memory中所指的resource lock住

// 若mark掉LockResource這行, 會產生 ”Jpeg Error #52” 的錯誤

  pRes := LockResource(hMem);
  // SizeOfResource取得指定resource的Byte數

    ResSize := SizeOfResource(hInstance, hRes);
    // 設定 memory stream 的大小

  MemStream.SetSize(ResSize);
// 將pRes指標所指的內容寫入MemStream中, 

// 寫入ResSize個byte數

    MemStream.Write(pRes^, ResSize);
  FreeResource(hMem); // Free掉hMem所指的resource

    // 將指標移回0的位置, 否則會產生 ”Jpeg Error #41” 的錯誤

    MemStream.Seek(0, 0);
JPEGImage.LoadFromStream(MemStream);

    Image1.Canvas.Draw(0, 0, JPEGImage);
  finally

  JPEGImage.Free;

    MemStream.Free;
  End;

End;

範例二:

var
    ResStream: TResourceStream;

    jpg: TJPegImage;

begin
  jpg := TJPegImage.Create;

  ResStream := TResourceStream.Create(hInstance, 'kaoru', RT_RCDATA);
  try

ResStream.Seek(0, soFromBeginning); // 指標移回ResStream的開頭

    jpg.LoadFromStream(ResStream); // 將 ResStream 載入 jpg 中

    Image1.Picture.Assign(jpg); // 將jpg顯示在Image1上

  finally

    jpg.Free;

    ResStream.Free;
  end;

end;
技巧提示
1.由於 Delphi 的 TImage 元件不支援 Jpeg, 故得借助 TJpegImage 類別的幫助, 使用時必須 uses Jpeg。
2.編輯 .rc 檔時, resource type欄位可用 RCDATA或JPEG。

1-5 使用resource中的ICON
使用元件 

TImage, TImageList, TComboBox

原理解說 

設計步驟 

1.將Icon的resource連結至專案中

2.範例一以LoadIcon API取得resource icon

3.範例二以ExtractIcon API取得resource icon

4.範例三以TResourceStream的方式取得resource icon

範例說明 

範例一:

LoadIcon的宣告如下,

HICON LoadIcon(

    HINSTANCE hInstance,
// handle of application instance

    LPCTSTR lpIconName 
// icon-name string or icon resource identifier

);
LoadIcon的使用方式與LoadBitmap的方式相同, 成功的話回傳resource內指定的Icon的handle。

procedure TForm1.btnLoadIconFromResClick(Sender: TObject);
begin
  // 這行將resource中的識別ID為1的Icon取代Form1的Icon

  Self.Icon.Handle := LoadIcon(hInstance, MAKEINTRESOURCE(1));
  // 這行將resource中的識別名識為car的Icon取代Application的Icon

  Application.Icon.Handle := LoadIcon(hInstance, PChar('car'));

end;
範例二:

ExtractIcon的宣告如下,

HICON ExtractIcon(

    HINSTANCE hInst,
// instance handle

    LPCTSTR lpszExeFileName,
// filename of file with icon

    UINT nIconIndex 
// index of icon to extract

   );
ExtractIcon可取得一module(Exe或DLL檔)內的icon。第一個參數hInstance指呼叫此API的Application的Instance handle; 第二個參數lpszExeFileName指要取得的icon所在的Exe/DLL或icon檔; 第三個參數nIconIndex指要取得的icon的index, 所此值為0, 表取得第一個icon的handle, 若此值為-1, 則此API回傳module內的icon總數。

此範例先以ExtractIcon取得專案內所有的icon總數, 再以ExtractIcon配合for迴圈, 取得專案內每個icon, 並丟給MyIcons陣列以及ImageList1元件, 此時將每個icon的index值加入ComboBox1的Items中, 最後一行將Application的Icon改成ImageList1元件中的第一個icon。使用者可在ComboBox1的OnChange Event中, 以ComboBox1的ItemIndex值來取得ImageList1相對應的icon, 並畫在Image1上。

procedure TForm1.btnExtractIconClick(Sender: TObject);

var 
i, iIconCount: integer;

  MyIcons: array of TIcon;

Begin
// 最後一個參數代入 -1 取得 project1.exe 內的Icon總數目

// 由於ExtractIcon的最後一個參數型態為UINT

// 故用Cardinal來轉型

  iIconCount := ExtractIcon(hInstance, PChar(ParamStr(0)), Cardinal(-1));
SetLength(MyIcons, iIconCount); // 設定MyIcons陣列長度

  For i := 0 to iIconCount - 1 do
begin
    MyIcons[i] := TIcon.Create;

    Myicons[i].Handle := Extracticon(hInstance, PChar(ParamStr(0)), i);
    // 將取得的新Icon加入ImageList1元件

    ImageList1.AddIcon(Myicons[i]); 
    ComboBox1.Items.Add(IntToStr(i));

  end;

  combobox1.ItemIndex := 0;
  // 將Application的Icon替換成Index為0的這個Icon

  ImageList1.GetIcon(0, Application.Icon);

end;
procedure TForm1.ComboBox1Change(Sender: TObject);
begin

  ImageList1.GetIcon(Combobox1.Itemindex, Image1.Picture.Icon);
end;
範例三:

重點在Create TResourceStream時, 有兩種方法, 一為TResourceStream.Create, 一為TResourceStream.CreateFromID, 前者傳入resource的識別名稱, 後者傳入resource的識別ID。

procedure TForm1.btnLoadCursorFromResStreamClick(Sender: TObject);
var 
ResStream1, ResStream2: TResourceStream;

Begin
  // TResourceStream.Create是以ResName為識別來取得所要之resource

  ResStream1 := TResourceStream.Create(hInstance, 'car', RT_ICON);

  try

    Application.Icon.LoadFromStream(ResStream1);

  finally

    ResStream1.Free;

  end;

  // TresourceStream.CreateFromID是以ResID為識別來取得所要之resource

  ResStream2 := TResourceStream.CreateFromID(hInstance, 1, RT_ICON);

  try

    Self.Icon.LoadFromStream(ResStream2);

  finally

    ResStream2.Free;

  end;

end;
技巧提示
1-6 使用resource中的Cursor(含Animated Cursor)
使用元件 

無

原理解說 

靜態Cursor的副檔名為 .cur, 動態Cursor的副檔為 .ani。使用resource中的靜態Curosr可用LoadCursor API; 動態Cursor則不像靜態Cursor能直接取用, 必須取出後存成一暫存檔方能使用。

TScreen的Cursors屬性是一個陣列, 它列出了Application所有可用的cursor, 其型態為HCursor。TScreen的Cursor屬性型態為TCursor, 是index, 其值由-32768至32767, 更改TScreen.Cursor值, 則在Application內的所有Form皆會使用此Cursor。若想更改在某個Control內的Cursor, 則以更改TControl.Cursor的值即可, 範例一則是更改Form1的Cursor。

設計步驟 

1.分別將靜態Cursor與動態Cursor加入專案的resource中

2.用LoadCursor API取用靜態的resource Cursor

3.用TResourceStream取用動態的resource Cursor

範例說明 

範例一:
LoadCursor的宣告如下,

HCURSOR LoadCursor(

    HINSTANCE hInstance,
// handle of application instance

    LPCTSTR lpCursorName // name string or cursor resource identifier

   );
用法為傳入module的Instance Handle與resource內的Cursor識別名稱或識別ID。範例一傳入Application的hInstance值, Cursor的識別名稱為pikacur。成功的話, 則回傳值為該resource cursor在module內的handle。
procedure TForm1.btnLoadCursorFromResClick(Sender: TObject);

const
 MyCursor = 1;

begin
  // 將Screen的Cursor陣列index值為1的Cursor

  // 置換為resource內Cursor識別名稱為pikacur的Cursor

  Screen.Cursors[MyCursor] := LoadCursor(hInstance, PChar('pikacur'));

  Cursor := MyCursor; // 並將Form1的Cursor置換為新的Cursor
end;
範例二:

用TResStream的方式取出專案內識別名稱為MyAni1的resource cursor後, 將其存成一個暫存檔, 檔名為TempFile.Ani。最後再以LoadImage API將Screen.Cursor改成暫存檔這個動態Cursor。

LoadImage的宣告如下,

HANDLE LoadImage(

   HINSTANCE hinst, // handle of the instance that contains the image

   LPCTSTR lpszName, // name or identifier of image

   UINT uType,
// type of image

   int cxDesired,
// desired width

   int cyDesired,
// desired height

   UINT fuLoad
// load flags

);
LoadImage可取得可執行module內的icon, cursor或bitmap。第一個參數傳入Application的Instance handle, 第二參數傳入暫存動態cursor檔的檔名, 第三個參數值IMAGE_CURSOR(因為要載入的型態是Cursor), 第四, 五個參數值為0, 最後一個參數值用LR_LOADFROMFILE。函式成功的話, 傳回值為所載入的Image的handle。

procedure TForm1.btnLoadCursorFromResClick(Sender: TObject);

const

  NewCursor = 1;

var

  ResStream: TResourceStream;

begin

  ResStream := TResourceStream.Create(hInstance, 'MyAni1', 'ANICURSOR');
  try

ResStream.SaveToFile('TempFile.ANI');

    Screen.Cursors[NewCursor] := LoadImage(hInstance,

      'TempFile.ANI', 
IMAGE_CURSOR, 0, 0, LR_LOADFROMFILE);

  Screen.Cursor := NewCursor;
finally

  ResStream.Free;

end;

end;
技巧提示
在Help中查詢TCursor即可得知預設的TScreen.Cursors陣列有哪些預設的Cursor可用, 更改Cursor值時可直接以指定數值的方式, 例: Screen.Cursor := 0; 的方式, 或用const的方式, 例: Screen.Cursor := crSQLWait。

1-7 使用resource中的AVI
使用元件 

TAnimate, TMediaPlayer

原理解說 

設計步驟 

1.將一 .avi檔連結至專案的resource中

2.利用TAnimate元件播放resource中的AVI

3.利用TResourceStream與TMediaPlayer元件播放resource中的AVI

範例說明 

範例一:
運用TAnimate元件播放 .avi檔動作非常簡單, 只要將AVI的resource識別名稱指定給TAnimate的ResName屬性(若為識別ID者可利用ResID屬性), 再將resource的的Instance handle指定給TAnimate元件的ResHandle屬性, 最後設TAnimate.Active屬性為True即可播放。
procedure TForm1.btnPlayAviFromResClick(Sender: TObject);

begin

  Animate1.ResName := 'dunk';

  Animate1.ResHandle := hInstance;

  Animate1.Active := True;

end;
範例二:

TAnimate元件有一嚴重的限制, 它只能播放沒有聲音的AVI, 若要播放有聲音的AVI, 可利用TMediaPlayer元件。但TMediaPlayer元件沒有像TAnimate元件的ResID/ResName這麼便利的屬性, 可以直接播放resource中的AVI, 解決方法是利用TResourceStream取出AVI的resource後, 先存一暫存成檔, 再將TMediaPlayer的FileName屬性指定為該暫存檔, 即可順利播放; 播放完畢後, 再將暫存檔刪除即可。

procedure TForm1.btnPlayAviFromResByMdeiaPlayerClick(Sender: TObject);

var

  ResStream: TResourceStream;

begin

  ResStream := TResourceStream.Create(hInstance, 'dunk', 'AVI');

  try

    ResStream.SaveToFile('TempAvi.avi');

    Mediaplayer1.FileName := 'TempAvi.avi';
    // Display屬性為指定播放時, 其輸出是利用哪個window

    // 本範例是將將其輸在Panel1元件上

    // 若Display屬性設為nil

    // 則播放時, TMediaPlayer元件會自己產生一個window

    Mediaplayer1.Display := Panel1;

    // Device Type屬性指定media player的device type屬性

    // 本範例要播放AVI檔, 其值設dtAVIVideo
    Mediaplayer1.DeviceType := dtAVIVideo;
    // 指定DeviceType屬性值後, 記得執行Open device的動作

    // 否則會出現No MCI device open的錯誤訊息

    Mediaplayer1.Open;

    MediaPlayer1.Play; // 開始播放

  finally

    ResStream.Free;

    DeleteFile('TempAvi.avi');

  end;

end;
技巧提示
1.TAnimate元件只能播放沒有聲音的 .avi檔, 否則會出現 "Cannot open AVI" 的錯誤訊息.

2.在編輯 .rc檔時, AVI檔的resource type欄位值須為AVI, 不可為RCDATA, 否則出現 "Cannot open AVI" 的錯誤訊息.

1-8 使用resource中的Wav
使用元件 

無

原理解說 

.wav是一種聲音檔, windows內有許多聲音檔可拿來使用, 如C:\Winnt\Media目錄下ringin.wav, ringout.wav等檔即是。

設計步驟 

1.將 .wav檔加入專案的resource中, 本節的範例利用windows內建的登入音效, 將其更名為login.wav後使用。

2.第一種方法, 利用FindResource/LoadResource/LockResource等API取出resource中的WAV後, 再以SndPlaySound函式播放

3.第二種方法, 直接利用PlaySound來播放resource中的WAV

範例說明 

範例一:

在使用resource中的Jpeg一節中, 已對FindResource/LoadResource及LockResource等API做了說明, 在此不再贅述。SndPlaySound函式的宣告如下,

BOOL sndPlaySound(

    LPCSTR lpszSound,


    UINT fuSound


   );
第一個lpszSound字串參數指定要播放的聲音, 這聲音可指registry或win.ini檔中所指定的聲音, 也可以是一個 .wav的聲音檔。第二個參數fuSound是SndPlaySound函式動作時的一個Flag, 其可能值如下,

SND_ASYNC: 聲音播放是不同步的, 且聲音一被播放時, 函式立刻傳回其傳回值。若要中斷不同步的聲音播放時, 須再度呼叫此函式, 且lpszSound參數值為nil。

SND_LOOP: 聲音重覆地一直播放, 直到SndPlaySound再度被呼叫,而且lpszSound的參數為nil時。

SND_MEMORY: 指要播放的聲音存在記憶體內。

SND_NODEFAULT: 如果找不到要播放的聲音時, SndPlaySound函式不播放系統預設的聲音。

SND_NOSTOP: 如果有個聲音正在播放, 函式馬上回傳False, 且不播放要播放的聲音。

SND_SYNC: 聲音播放是同步的, 且函式要等到聲音播放完畢時才會回傳回傳值。

函式播放聲音成功時, 回傳值為True, 否則回傳False。

最後記得要 uses MMSystem, 因為SndPlaySound這個函式宣告在MMSystem這個unit。

uses MMSystem;

procedure TForm1.btnPlayWaveFromResClick(Sender: TObject);

var

  hFind: THandle;

hRes: THandle;

  pWave: PChar;

begin

  hFind := FindResource(HInstance, PChar('login'), PChar('WAVE'));

  if hFind <> 0 then begin

    hRes := LoadResource(HInstance, hFind);

    if hRes <> 0 then begin

      pWave := LockResource(hRes);

      if Assigned(pWave) then

        SndPlaySound(pWave, snd_ASync or snd_Memory);

      UnlockResource(hRes);

    end;

    FreeResource(hFind);

  end;

end;
範例二:

範例二較為簡單, 直接利用PlaySound來播放resource中的WAV(記得uses MMSytesm)。PlaySound的宣告如下,

BOOL PlaySound(

    LPCSTR pszSound,


    HMODULE hmod,


    DWORD fdwSound


   );


第一個參數pszSound指定要播放的聲音; 第二個參數hmod指擁有該聲音的moudle handle; 第三個參數fdwSound為播放聲音的flag, 其可能值參考win32的help, 本範例是要播放resource中的WAV, 所以使用SND_RESOURCE或SND_MEMORY這個值, 而SND_SYNC flag指同步的播放聲音, 播放完畢時, 函式才回傳回傳值。

uses MMSystem;

procedure TForm1.btnPlayWaveFromResClick(Sender: TObject);

begin

  PlaySound(Pchar(‘login’), hInstance, SND_RESOURCE or SND_SYNC);

  // 或是PlaySound(Pchar(‘login’), hInstance, SND_MEMORY or SND_SYNC);

end;
技巧提示
1.其它的多媒體檔(如mp3、midi等)亦可使用同樣的方式加入專案的resource中, 利用範例二介紹的TResourceStream的方式, 即可使用。

1-9 使用resource中的Font
使用元件 

無

原理解說 

設計步驟 

1.將所需的字型檔加入專案的resource中

2.利用TResourceStream將取出resource中的Font。
3.以AddFontResource API將取出的Font加入系統的Font Table中, 再以SendMessage的方式通知其它windows說系統的Font Table已有異動。
4.程式關閉時, 以RemoveFontResource將Font自系統中移除。
範例說明 

首先以TResourceStream的Create或CreateFromID的方式將resource中的Font取出, 取出後須將它存成一檔案才能使用, 儲存路徑建議可存在系統路徑下的Fonts目錄。接著呼叫AddFontResource將新取出的Font加入Windows Font Table中, AddFontResource API宣告如下,

int AddFontResource(

    LPCTSTR lpszFilename // pointer to font-resource filename 

   );
lpszFilename參數值為一有效的字型檔案名稱, 有效的字型檔副檔名可能為 .FON、.FNT、.TTF或 .FOT。函式成功的話, 其回傳值為大於0之整數值, 失敗則回傳值為0。至此為止還無法利用新加入的字型, 須呼叫SendMessage API, 通知所有的top-level windows說Windows Font Table已變更後, 才能使用新加入的字型。SendMessage的宣告如下,

LRESULT SendMessage(

    HWND hWnd,
// handle of destination window

    UINT Msg,
// message to send

    WPARAM wParam,
// first message parameter

    LPARAM lParam 
// second message parameter

   );
第一個參數hWnd表接收Message的目標視窗handle, 在此範例使用廣播的方式, 即代入HWND_BROADCAST; 第二參數Msg表要傳送的Message, 這裏使用WM_FONTCHANGE即可。

接下來即可使用新字型, 方式即更改TControl.Font屬性值即可。

procedure TForm1.btnUseFontFromResClick(Sender: TObject);

var

  ResStream: TResourceStream;
  SysDir: array[0..MAX_PATH-1] of Char;

begin

  ResStream := TResourceStream.Create(hInstance, 'Lsansuni', PChar('MYFONT'));
  try

GetSystemDirectory(SysDir, MAX_PATH); // 取得系統目錄

IncludeTrailingBackSlash(SysDir);

  ResStream.SavetoFile(SysDir + 'Lsansuni.ttf');
finally

  ResStream.Free;
end;

  if AddFontResource(PChar(SysDir + 'Lsansuni.ttf')) = 0 then begin

    ShowMessage(SysErrorMessage(GetLastError));

    Exit;

  end;
  SendMessage(HWND_BROADCAST, WM_FONTCHANGE, 0, 0);

  // 將 TButton的字型改為新加入的字型, 且大小設為12

  btnUseFontFromRes.Font.Name := 'Lsansuni';

  btnUseFontFromRes.Font.Size := 12;

end;
若不再使用新加入的字型, 欲將其移除時, 先將字型檔刪除, 再執行RemoveFontResource API將新字型自Windows Font Table中移除, RemoveFontResource用法與AddFontResource相同, 傳入欲移除的字型檔名稱即可, 最後執行SendMessage通知其它windows說Windows Font Table已變更。

begin
  DeleteFile(SysDir + 'Lsansuni.ttf'));

  RemoveFontResource(PChar(SysDir + 'Lsansuni.ttf'));

  SendMessage(HWND_BROADCAST, WM_FONTCHANGE, 0, 0);

end;

技巧提示
1-10 使用resource中的Html
使用元件 

TWebBrowser

原理解說 

HTML Pages(包含相關的檔案, 如gif、bmp或Style Sheet等)也可以加入專案的resource中, 然後在Delphi內用TWebBrowser元件搭配res protocol即可顯示resource中的網頁。將HTML Pages加入resource的好處有: 使用者無法檢視網頁原始碼、不會將網頁儲在硬碟的暫存目錄中等。

res Protocol 如同http、ftp, 是一種自module中取得特定resource的通訊協定, 適用Microsoft IE 4.0以後的版本, 其用法為: res://sFile[/sType]/sID, 其中sFile指包含resource的完整檔名; sType指數字或字串的resource type; sID則指數字或字串的resource ID。

設計步驟 

1.編輯所需的HTML檔, 其內容如下:

<html>

<head>

<title>use HTML from resource in Delphi</title>

</head>

<body bgcolor=black>

  <p align=center>

  <b><font color=red size=5>using HTML from Delphi resource:</font></b>

  </p>
  <img src="HtmlGif">

</body>

</html>
注意<img src="HtmlGif">這行, 標準的語法應當是<img src="img檔案名稱">, 這裏則須使用resource的識別名稱。
2.將HTML網頁加入專案的resource中

3.利用TWebBrowser元件與res protocol來顯示resource中的網頁。

範例說明 

首先利用res protocol的語法(res://sFile[/sType]/sID), 組合成一個URL字串, 再傳入TWebBrowser.Navigate的URL參數中。檢視resource中網頁的範例程式如下:

procedure TForm1.btnUseHtmlFromResClick(Sender: TObject);

var

  sURL: string;

begin

  sURL := 'res://' + Application.ExeName + '/TestHtml';
  // 將ResName為TestHtml的resource顯示在WebBrowser1上

  WebBrowser1.Navigate(sURL);

end;
技巧提示

1-11 使用resource中的String Table
使用元件 

無

原理解說 

設計步驟 

1.將所要利用的Sring Tble加入專案的resource中, String Table的 .rc檔格式與之前所述不同, 其格式如下:

STRINGTABLE

 {

  101, "This is string 1"

  102, "This is string 2"

  103, "This is string 3"

 }
或以下的格式也可以:

STRINGTABLE

BEGIN

  101, "This is string 1"

  102, "This is string 2"

  103, "This is string 3"
END

其中的101, 102, 103即為各個string的識別ID。

2.分別利用LoadStr與LoadString函式取出resource中的string。

範例說明 

範例一:

利用LoadStr函式取出resource中的string, 其方式非常簡單, 只要傳入string的識別ID即可。LoadStr函式的宣告如下,

function LoadStr(Ident: Integer): string;
procedure TForm1.btnLoadStrClick(Sender: TObject);

begin

  ShowMessage(LoadStr(101)); 

  ShowMessage(LoadStr(102));

  ShowMessage(LoadStr(103));

end;
LoadStr是為了之前版本的delphi而保留, 其功能被resourcestring所取代。resourcestring的使用方式就如同宣告const一樣, 其範例如下,

resourcestring

  CreateError = 'Cannot create file %s';

  OpenError = 'Cannot open file %s'; 

  LineTooLong = 'Line too long';

  ProductName = 'Borland Delphi\000\000';

被宣告成resourcestring的string會被delphi當作resources, 這些string可以被修改且不用再編譯程式, 所以可用來製作多國語言的程式。

範例二:

利用LoadString API取得resource中的string, 其宣告如下,

int LoadString(

    HINSTANCE hInstance, // handle of module containing string resource 

    UINT uID,
// resource identifier 

    LPTSTR lpBuffer, // address of buffer for resource 

    int nBufferMax 
// size of buffer 

   );
第一個參數hInstance指擁有string resource的module的Instance handle, 第二個參數uID指string resource的識別ID, 範例中的101,102,103等數值即是, 第三個參數為取得string resource的buffer, 第四個參數為第三個參數的size值。使用範例程式如下,

procedure TForm1.btnLoadStringClick(Sender: TObject); 
var 
  buffer: array[0..255] of Char; 
begin 
  Loadstring(hinstance, 101, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
  Loadstring(hinstance, 102, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
  Loadstring(hinstance, 103, @buffer, 255); 
  ShowMessage(StrPas(buffer)); 
end;
技巧提示
1-12 使用resource中的RTF, 並載入TRichEdit
使用元件 

TRichEdit

原理解說 

設計步驟 

1.先利用TRichEdit元件, 產生一個 .rtf檔, 存檔後, 依前述方式, 加入專案的resource中。

2.範例一以TResourceStream的方式取得resource中的RTF

3.範例二以FindResource/LoadResource/LockResource等API, 配合TStringList取得resource中的RTF

範例說明 

範例一:

直接以TResourceStream的Create/CreateFormID將resource中的RTF載入Stream中, 再利用TRichEdit的Lines.LoadFromStream即可完成。
procedure TForm1.btnLoadRTFFromResClick(Sender: TObject);

var ResStream: TResourceStream;

begin

  ResStream := TResourceStream.Create(hInstance, 'MyRtf', PChar('RTF'));

  try

    RichEdit1.Lines.LoadFromStream(ResStream);

  finally

    ResStream.Free;

  end;

end
範例二:

利用FindResource/LoadResource/LockResource等API將resource內的RTF的載入memory, 並取得其指標後, 以TString的SetText方法將其載入, 然後存成一暫存檔。最後以TRichEdit的Lines.LoadFromFile將此暫存檔載入即可, 檔案載入後, 便將該暫存檔刪除。
procedure TForm1.Button1Click(Sender: TObject);

const

  TmpFile = 'temp.rtf';

var

  hFind: THandle;

hRes: THandle;

  pData: PChar;

  sl: TStringList;

begin
  hFind := FindResource(hInstance, 'MyRtf', 'RTF');

if hFind = 0 then 
begin
ShowMessage(‘Could not find resource’);

Exit;

end;
hRes := LoadResource(hInstance, hFind);
if hRes <> 0 then 
begin

  pData := LockResource(hRes);

  sl := TStringList.Create;

  try

    sl.SetText(pData);

     sl.SaveToFile(TmpFile);

  finally

    sl.Free;

    UnlockResource(hRes);

    FreeResource(hRes);

  end;

    RichEdit1.Lines.LoadFromFile(TmpFile);

    DeleteFile(TmpFile);

 end;
end;
技巧提示

1.範例二若直接使用RichEdit1.Lines.SetText(pData)的方式來載入RTF的話, 資料可載入, 但內容全部是亂碼, 它無法載換成正確的格式來顯示, 故先儲存為一暫存檔後再載入RichEdit1。

2.使用者可以相同方法將其它檔案加入resource中並使用。
1-13 產生有Version Info的resource
使用元件 

無

原理解說 

在Delphi3之後的版本, 要在專案中加入Version Info變得非常簡單, 使用者只須點選Project -> Options選項中的Version Info頁, 將Include version information in project核取方塊打勾, 然後在Module version number、Module attributes、Language及各Key的Value填入值即可。
以下介紹如何在Delphi2的專案中產生Version Info。

設計步驟 

1.先產生一個空白的 .rc檔, 其內容如下(以Delphi5中的delphi32.exe為範例):

1 VERSIONINFO

FILEVERSION 5,0,6,18

PRODUCTVERSION 5,0,0,0

FILEOS 0x10004

FILETYPE 0x1

{

BLOCK "StringFileInfo"

{


BLOCK "040904E4"


{



VALUE "CompanyName", "Inprise Corporation"



VALUE "FileDescription", "Delphi-32 Development Environment"



VALUE "FileVersion", "5.0.6.18"



VALUE "InternalName", "DELPHI32"



VALUE "LegalCopyright", "Copyright c 1996-1999 Inprise Corporation"



VALUE "OriginalFilename", "DELPHI32.EXE"



VALUE "ProductVersion", "5.0"



VALUE "ProductTitle", "Update Pack 1"



VALUE "ProductName", "Enterprise"


}

}

BLOCK "VarFileInfo"

{


     VALUE "Translation", 0x0409 0x04E4

}

}

2.存檔, 檔名為MyVersionInfo.rc。

3.利用BRCC32.Exe, 將MyVersionInfo.rc轉換為MyVersionInfo.res。

範例說明 

VersionInfo的格式為: 

versionID  VERSIONINFO  fixed-info  { block-statement . . . }
MyVersionInfo.rc內容說明如下:

1 VERSIONINFO:這行宣告它是VERSIONINFO的resource, 其中1指versionID, 它的值必須是1.

fixed-info: 包括FileVersion, Product Version, FileOS, FileType等, 說明如下:

	Statement
	Description

	FILEVERSION
	定義該file的二進位版本, 由4個16-bit的整數組成, 第一個數字為Major version, 第二個數字表Minor version, 第三個數字表Release¸第四個數字表Build

	PRODUCTVERSION
	定義product該product的二進位版本, 由4個16-bit的整數組成.

	FILEOS
	指該file設計於何種OS平台, 範例中的值VOS__Windows32是指為32-bit windows所設計

	FILETYPE
	指該file的type, 範例中的值VFT_APP指該file是一個Application.


StringFileInfo Block: 指定一個以上的版本資訊區塊. 其格式如下:

BLOCK "StringFileInfo"  
{
 BLOCK "lang-charset"
 {          
VALUE "string" "string value"
}

}

lang-charset Block: 0409指langID, 其值為U.S. English; 04E4指charset.

而string-name包含如下:

	Name
	Description

	CompanyName
	指生產此file的Company, 此欄位為必須

	FileDescription
	對該file的描述, 此欄位為必須

	FileVersion
	該file的版本數字, 此欄位為必須

	InternalName
	如果file是一個DLL, InterName指一個module name; 若該file沒有internal name, 則此欄位值為原本的file name. 此欄位為必須

	LegalCopyRight
	適用此file的版權宣告

	OriginalFileName
	該file的original name, 不包含路徑. 這項資訊可判斷使用者是否更改了file的name. 此欄位為必須

	ProductName
	與file一起分發出去的product的name. 此欄位為必須

	ProductVersion
	上述product的版本. 此欄位為必須

	ProductTitle
	product的抬頭名稱


技巧提示

1-14 取得resource中的Version Info
使用元件 

TMemo

原理解說 

取得resource中的VersionInfo其方法與之前用過的不同, 必須利用GetFileVersionInfoSize/GetFileVersionInfo/VerQueryValue這三個API。

設計步驟 

範例說明 

首先執行GetFileVersionInfoSize API, 判斷一個module是否含有版本資訊以及判斷容納檔案版本資訊需要一個多大的緩衝區, 如果函式成功的話, 則回傳版本資訊的大小byte數, 失敗則回傳0。GetFileVersionInfoSize的宣告如下,

DWORD GetFileVersionInfoSize(

    LPTSTR lptstrFilename,
// pointer to filename string

    LPDWORD lpdwHandle // pointer to variable to receive zero

   );
第一個字串參數lptstrFilename指要取得版本資訊的檔案名稱, 第二個參數lpdwHandle為一個接收函式回傳0的變數。

取得版本資訊的大小後, 再執行GetFileVersionInfo API, 這個函數會檢查指定檔案內的版本資訊﹐並將需要的資料複製到一個緩衝區裏, 其宣告如下,

BOOL GetFileVersionInfo(

    LPTSTR lptstrFilename,
// pointer to filename string

    DWORD dwHandle,
// ignored 

    DWORD dwLen,
// size of buffer

    LPVOID lpData 
// pointer to buffer to receive file-version info.
第一個字串參數lptstrFileName指要取得版本資訊的檔案名稱, 第二個參數dwHandle在win32下未使用, 第四個參數指定一個位元組緩衝區的第一個位元組, 該緩衝區用於裝載檔案的版本資訊, 第三個參數dwLen指由lpData參數指定的位元組陣列或緩衝區的大小。函式成功後回傳值為非0, 我們要的東西就在buffer變數中, 此時便可利用VerQueryValue API來取得buffer變數中的各項資訊。VerQueryValue API的宣告如下,

BOOL VerQueryValue(

    const LPVOID pBlock,
// address of buffer for version resource

    LPTSTR lpSubBlock,
// address of value to retrieve

    LPVOID *lplpBuffer,
// address of buffer for version pointer

    PUINT puLen 
// address of version-value length buffer

   );
第一個參數pBlock指定一個記憶體塊第一個位元組的位址, 這個記憶體塊包含了由GetFileVersionInfoapi取回的版本資訊; 第二個參數lpSubBlock指要讀取哪個sub-block的版本資訊, 它是由 \ 所組成, 其三個可能值如下,

	Form
	Description

	\
	獲取檔案的VS_FIXEDFILEINFO架構

	\VarFileInfo\Translation
	獲取檔案的翻譯表

	\StringFileInfo\lang-charset\string-name
	獲取檔案的字串資訊(參考上一節VersionInfo的格式部份)


第三個參數lplpBuffer為一個變數的位址﹐該變數用於裝載一個緩衝區的位址, 要求的版本資訊最終會裝載到那個緩衝區裏; 第四個參數puLen指定由lplpBuffer參數引用的數據值長度。
範例程式如下:

procedure TForm1.btnReadVersionInfoFromResClick(Sender: TObject);

var

  VerSize: DWORD;

  Zero: THandle;

  pBlock: Pointer;
  sSubBlock: string;

  pBuffer: Pointer;

  Size: UINT;

begin

  // 先取得VersionInfo resource的大小

  VerSize := GetFileVersionInfoSize(PChar(ParamStr(0)), Zero);

  if VerSize <> 0 Then // 成功取得VersionInfo resource的大小

  begin

    GetMem(pBlock, VerSize); // 配置 memory

    try

  // 取得VersionInfo resource, 並將資料裝載於pBlock變數中

      GetFileVersionInfo(PChar(ParamStr(0)), 0, VerSize, pBlock);
    GetMem(pBuffer, 256);
    try

    // sub-block的組成

// 1.StringFileInfo

      // 2.lang-charset, 本範值為lang: 0409; charset: 04E4

        //  lang與charset皆為16進位數字

        // 3.string-name, 底下這行的string-name值為CompanyName

        sSubBlock := '\\StringFileInfo\\040904E4\\CompanyName';
        // 以VerQueryValue取得pBlock中的pSubBlock的版本資訊

        VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

        Memo1.Lines.Add('Company Name: ' + StrPas(pBuffer));
        sSubBLock := '\\StringFileInfo\\040904E4\\FileDescription';
        VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

        Memo1.Lines.Add('File Description: ' + StrPas(pBuffer));
        sSubBLock := '\\StringFileInfo\\040904E4\\FileVersion';
        VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

        Memo1.Lines.Add('File Version: ' + StrPas(pBuffer));
        sSubBLock := '\\StringFileInfo\\040904E4\\ProductName';
        VerQueryValue(pBlock, PChar(sSubBlock), pBuffer, Size);

        Memo1.Lines.Add('Product Name: ' + StrPas(pBuffer));
      finally

        FreeMem(pBuffer);

      end;

    finally

      FreeMem(pBlock);

    end;

  end;
end;
技巧提示

1-15產生及利用resource DLL
使用元件 

無

原理解說 

resource除了可加入EXE執行檔中, 亦可加入DLL檔中, 其方式與加入EXE執行檔的方式大同小異。

設計步驟 

在Delphi主選單, 選New --> DLL, Delphi便自動產生一個library專案。再將之前產生的sample.res檔連結至專案後存檔, 方法為在 {$R *.RES} 這行下方加入 {$R sample.RES} 一行, 檔名為ResTest.dpr. 編譯後在專案目錄下即產生一名為ResTest.DLL的檔案, 此DLL檔即是我們要的resource DLL。專案內容如下:

library ResTest;

{ Important note about DLL memory management: ShareMem must be the

  first unit in your library's USES clause AND your project's (select

  Project-View Source) USES clause if your DLL exports any procedures or

  functions that pass strings as parameters or function results. This

  applies to all strings passed to and from your DLL--even those that

  are nested in records and classes. ShareMem is the interface unit to

  the BORLNDMM.DLL shared memory manager, which must be deployed along

  with your DLL. To avoid using BORLNDMM.DLL, pass string information

  using PChar or ShortString parameters. }

uses

  SysUtils,

  Classes;

{$R *.RES}

{$R sample.RES} // 加入這一行, 將sample.res連結至專案

begin

end.
範例說明 

本範例示範何如何將Application的Icon置換為ResTest.DLL中的resource Icon。首先利用LoadLibrary API將ResTest.DLL檔載入, 再利用之前提過的LoadIcon函數取得resource識別名稱為car的Icon, 指定給Application.Icon.Handle即可。使用完畢, 記得執行FreeLibrary API來釋放載入的ResTest.DLL。

procedure TForm1.Button1Click(Sender: TObject);

var

  hLib: THandle; 

  hIcon: THandle;

begin

  hLib := LoadLibrary('C:\Test\RESTEST.DLL');
  try

  if h <= 32 then

      ShowMessage(SysErrorMessage(GetLastError))

    else 
begin

      hIcon := LoadIcon(hLib, 'car');
      Application.Icon.Handle := hIcon;

    end;

  finally
    FreeLibrary(hLib);
  end;

end;
技巧提示

�1-1 資源檔(.res)的產生與連結��1-2 列舉一module內所有resource�


1-3 使用resource中的Bitmap





1-4 使用resource中的Jpeg





1-5 使用resource中的Icon





6使用resource中的Cursor(含Animated Cursor)


1-7 使用resource中的AVI�1-8 使用resource中的WAV


1-9 使用resource中的Font�1-10使用resource中的HTML


1-11 使用resource中的string table�1-12 使用resource中的RTF, 並載入TRichEdit�1-13 產生有VersionInfo 的resource�1-14 取得resource中的VersionInfo�1-15 產生及利用resource DLL

